
Morphing Tilings of the Plane into Tilings of Surfaces 
 

Mark L. Loyola 
mloyola@ateneo.edu 

Department of Mathematics 
Ateneo de Manila University 

Philippines 
 

Ma. Louise Antonette N. De Las Peñas 
mdelaspenas@ateneo.edu 

Department of Mathematics 
Ateneo de Manila University 

Philippines 
 
Abstract:   This work discusses a procedure to generate a tiling 𝒯ௌ೑

 of a 2-dimensional surface 𝑆௙  embedded in the 

Euclidean 3-space ℝଷ from a tiling 𝒯 of the Euclidean plane ℝଶ. We employ the computer algebra system Mathematica 
to generate 3D graphical images of 𝒯 and 𝒯ௌ೑

 and render animations which create the effect of 𝒯 morphing into 𝒯ௌ೑
.  

      
1.  Introduction 
      

Tilings of surfaces, space forms, or quotient spaces have been interesting subjects of study in 
discrete geometry and computer graphics [10, 13, 14, 15]. These geometric structures find 
applications as artworks or as mathematical models of chemical structures such as analogs of carbon 
nanotubes [6] or carbon nanotori [11, 12]. 

In recent years, technological tools have played an important role in the teaching and research of 
tilings. Dynamic geometry software, for example, have allowed for the investigation of properties 
and relationships of tilings [3], and the discovery of new families of tilings [2, 5].  Computer algebra 
systems have facilitated an efficient way of rendering tilings in surfaces and quotient spaces [7]. 

One of the important benefits of a technological tool is being able to facilitate the visualization of 
geometric representations. Geometric concepts can be visualized with dynamic geometric 
representations such as graphical images, animations, videos, applets, or with constructed or drawn 
representations made with a computer program [18]. Interactive applets, for example, can facilitate 
the deeper understanding of otherwise static geometric images, as applets allow users to explore, 
conjecture, make discoveries on these geometric figures, and even arrive at mathematical proofs. 
Computer programs can produce visual outputs, such as visualization of tilings of higher dimensions 
and other spaces, facilitating mathematical modeling and simulation of real-life scenarios [4].  

In this paper, we present our work on tilings of surfaces, which are realized as geometric 
representations with the use of technology. We organize the paper as follows. In the next section, we 
enumerate different parametric systems of equations that define surfaces embedded in the Euclidean 
3-space and discuss a procedure to generate a tiling of a given surface from a tiling of the Euclidean 
plane. In Section 3, we implement the said procedure in a computer algebra system to generate 3D 
graphical images of such tilings. We present two different ways of implementing this procedure and 
discuss the programming code on how each can be accomplished. In the final section, we demonstrate 
a technique to produce animations with a slider and animated gif files that show a tiling of the plane 
transforming into a tiling of a surface.  

 



2.  Tilings of Surfaces 
 

A tiling or tessellation 𝒯 of a point space 𝕏 is a collection of subsets of 𝕏 called tiles whose 
interiors do not intersect and whose union is the entire space 𝕏. Although 𝕏 can, in principle, be any 
set, it is usually endowed with a geometric structure. Various examples of tilings of the Euclidean 
plane ℝଶ and their classifications in terms of symmetry and topological considerations are found in 
the classic text Tilings and Patterns [9]. One may also consult the searchable online databases Tiling 
Search [16] and Tilings Encyclopedia [8] to download images of tilings and retrieve useful 
information about them.  

Tilings with rich mathematical structures and at the same time aesthetically pleasing are those 
consisting of tiles which either belong to a single congruence or similarity class or to a finite number 
of such classes. We present in Figure 2.1 examples of tilings of two point spaces. Figure 2.1(a) depicts 
a square patch of the famous Ammann-Beenker tiling 𝒜 of ℝଶ [8]. This tiling consists of congruent 
squares and rhombi which are arranged around the center of the plane to form a pattern with an 8-
fold rotation symmetry. A star-shaped patch of an analog of this tiling on the unit 2-sphere 𝕊ଶ is 
shown in Figures 2.1(b) [front view] and 2.1(c) [back view]. 

 

   
(a) (b) (c) 

Figure 2.1 Amman-Beenker tilings of the point spaces (a) ℝଶ and (b, c) 𝕊ଶ. 
 

In this work, we shall restrict ourselves to point spaces 𝕏 that correspond to 2-dimensional 
surfaces embedded or lying in the Euclidean 3-space ℝଷ. Formally, we define a surface 𝑆௙ to be the 
image of a continuous map 𝑓 ∶ ℝଶ → ℝଷ in two variables 𝑢 and 𝑣. If we denote by (𝑥, 𝑦, 𝑧) the image 
in  ℝଷ of a point (𝑢, 𝑣) in ℝଶ, then 𝑥, 𝑦, 𝑧 may be identified with three continuous maps 𝑓௫, 𝑓௬, 𝑓௭, 
respectively, of 𝑢 and 𝑣. These identifications make 𝑆௙ a parametric surface and allow us to describe 
the surface conveniently using the parametric system of equations 

 
𝑥 = 𝑓௫(𝑢, 𝑣) 
𝑦 = 𝑓௬(𝑢, 𝑣) 
𝑧 = 𝑓௭(𝑢, 𝑣) 

(2.1) 

 
with parameters 𝑢, 𝑣 ∈ ℝ. For example, if we take the maps 𝑓௫(𝑢, 𝑣) = 𝑢, 𝑓௬(𝑢, 𝑣) = 𝑣, 𝑓௭(𝑢, 𝑣) =

0, we obtain the standard embedding of the Euclidean plane ℝଶ into ℝଷ. This trivially implies that 
the Euclidean plane may also be viewed as a surface. As a matter of fact, the point space 𝕊ଶ, which 
is the image of ℝଶ under the inverse stereographic projection, is also a surface. Observe that under 



this map, which is an inversion relative to 𝕊ଶ, the image of the small square patch of 𝒜 in Figure 
2.1(a) covers a large portion of the sphere as shown in Figure 2.1(b, c). The parametric equations 
defining these two and other surfaces, which we shall encounter later, are listed in Table 2.1 below. 
 
Table 2.1 Surfaces defined by a parametric system of equations with parameters 𝑢, 𝑣 ∈ ℝ, where 
𝑎, 𝑏 > 0 are constants and 𝜃௨,௩ ∈ [0, 2𝜋] is the angle measured in the counterclockwise direction 
between the vectors (1, 0) and (𝑢, 𝑣). 

Surface 𝑓௫ 𝑓௬ 𝑓௭ 

Plane ℝଶ 𝑢 𝑣 0 

Unit 2-Sphere 𝕊ଶ 
2𝑢

1 + 𝑢ଶ + 𝑣ଶ
 

2𝑣

1 + 𝑢ଶ + 𝑣ଶ
 

−1 + 𝑢ଶ + 𝑣ଶ

1 + 𝑢ଶ + 𝑣ଶ
 

Egg Carton ℂ௘ 𝑢 𝑣 sin 𝑢 + cos 𝑣 

Cone ℂ௢ 𝑎√𝑢ଶ + 𝑣ଶ

2𝜋
cos ൬

𝜃௨,௩

𝑎
2𝜋൰ 

𝑎√𝑢ଶ + 𝑣ଶ

2𝜋
sin ൬

𝜃௨,௩

𝑎
2𝜋൰ ඥ𝑢ଶ + 𝑣ଶ 

Cylinder ℂ௟ 
𝑎

2𝜋
cos ቀ

𝑢

𝑎
2𝜋ቁ 

𝑎

2𝜋
sin ቀ

𝑢

𝑎
2𝜋ቁ 𝑣 

Torus 𝕋 
𝑏 cos ቀ

𝑣
𝑏

2𝜋ቁ

√𝑎ଶ + 𝑏ଶ − 𝑎 cos ቀ
𝑢
𝑎 2𝜋ቁ

 
𝑏 sin ቀ

𝑣
𝑏

2𝜋ቁ

√𝑎ଶ + 𝑏ଶ − 𝑎 cos ቀ
𝑢
𝑎 2𝜋ቁ

 
𝑎 cos ቀ

𝑢
𝑎

2𝜋ቁ

√𝑎ଶ + 𝑏ଶ − 𝑎 cos ቀ
𝑢
𝑎 2𝜋ቁ

 

 
Given a tiling 𝒯 of ℝଶ and a parametric surface 𝑆௙, the collection 𝒯ௌ೑

≔ {𝑓(𝓉) ∣  𝓉 ∈ 𝒯} of images 

of tiles in 𝒯 forms a cover for 𝑆௙. If, in addition, the interiors of these images do not intersect, which 
depends ultimately on the map 𝑓, then 𝒯ௌ೑

 becomes a tiling of 𝑆௙. This happens, for instance, when 

𝑓 is a homeomorphism between ℝଶ and its image 𝑓(ℝଶ). Such is the case with the maps that define 
the surfaces ℝଶ, 𝕊ଶ, and ℂ௘ in Table 2.1.  
 
Illustration 2.1 We illustrate in Figure 2.2(a) the tiling of the egg carton surface ℂ௘ arising from 𝒜. 

 

  
 

(a) (b) 
Figure 2.2 (a) Tiling 𝒜ℂ೐

 of ℂ௘ arising from 𝒜. (b) The image in ℂ௘ of a rhombic tile in 𝒜. 



In Figure 2.2(b), eight selected vertices of 𝒜 are marked and connected by dashed line segments with 
their images in ℂ௘. Notice that the marked rhombic tile of 𝒜 underwent a deformation after it was 
mapped to ℂ௘. This is because the map used to define ℂ௘ is neither length-preserving (isometric) nor 
angle-preserving (conformal). Contrast this with the tiling 𝒜𝕊మ  in Figure 2.1(b) in which the angles 
(but not the lengths) of the tiles are preserved. This is a consequence of the fact that the inverse 
stereographic projection is conformal [1].  

The maps in Table 2.1 may be broadly classified as either injective or non-injective. As implied 
earlier, the first three belong to the first class. On the other hand, the presence of the periodic sine 
and cosine functions in the formulas for 𝑓௫ and 𝑓௬ when 𝑓 defines either a cone ℂ௢, a cylinder ℂ௬, or 
a torus 𝕋 hints at non-injectivity. Although this classification may seem purely algebraic at first 
glance, it has an important geometric implication when it comes to the construction of tilings of 
surfaces.  

The cone, cylinder, and torus maps, and hence the surface they define as well, are compatible only 
with tilings of ℝଶ that satisfy some symmetry properties. More particularly, a tiling  𝒯 of ℝଶ gives 
rise to a tiling of 

 

 ℂ௢ if 𝒯 possesses a 
ଶగ

௔
-fold rotation symmetry 𝑟௔ about the origin;  

 ℂ௟  if 𝒯 possesses a horizontal translation symmetry 𝑡௫ with vector of length 𝑎; and 
 𝕋 if 𝒯 possesses both a horizontal and a vertical translation symmetry 𝑡௫ and 𝑡௬ with vectors 

of lengths 𝑎 and 𝑏, respectively. 
 
The effect of any of these non-injective maps is to identify points in ℝଶ that are equivalent with 
respect to the group of symmetries generated by 𝑟௔, 𝑡௫, or 𝑡௫ and 𝑡௬. Identified points are then “glued” 
together to become a single point in the resulting surface, which may now be referred to as a space 
form or a quotient space. Consequently, identified tiles of the plane become a single tile under the 
map. We discuss a specific example in the next illustration.   
 
Illustration 2.2 Since the Ammann-Beenker tiling 𝒜 remains invariant under the 2-fold rotation 𝑟ଶ  
 

  
(a) (b) 

Figure 2.3 (a) Points in ℝଶ are identified with respect to 𝑟ଶ to obtain (b) a tiling of ℂ௢.       
 



about its center, which we assume to be the origin 𝑂, we can map it to the cone ℂ௢ by setting 𝑎 ∶= 2. 
Under this map, points in ℝଶ that are sent to each other by the rotation 𝑟ଶ are considered equivalent 
and get identified as a single point in the resulting cone. In Figure 2.3(a), vertices of 𝒜 that share the 
same color become a single vertex in the resulting tiling of the cone. This process may be visualized 
as the following sequence of actions: cut the square patch in Figure 2.3(a) along the dashed segment 
from a corner point up to the center to form two flaps; hold one of the flaps at the corner point; and 
then rotate it towards the opposite corner point so that vertices of the same color are glued together 
to form the cone whose apex is 𝑂 (see Figure 2.3(b)). Observe that, as expected, points such as the 
opposite corner points in Figure 2.3(a) are sent to the same points of ℂ௢. In addition, the angles of 
the tiles in 𝒜 are preserved in 𝒜ℂ೚

. This is because the cone map together with the cylinder and torus 
maps is conformal. We note, however, that the latter two maps are incompatible with 𝒜 precisely 
because this tiling is non-periodic [8]. That is, 𝒜 does not admit translation symmetries in its 
symmetry group. 
      
3.  Generating 3D Images of Tilings of Surfaces 
 

Our main objective in this and the next section is to illustrate a generic procedure to generate a 3D 
image of a tiling 𝒯ௌ೑

 of a surface 𝑆௙ from a given tiling 𝒯 of ℝଶ and render a simple animation that 

shows 𝒯 morphing into 𝒯ௌ೑
. We shall implement this procedure in Wolfram Mathematica [16], a 

powerful computer algebra system (CAS) developed by Wolfram Research. We take advantage of 
this system’s functional programming paradigm and its fast 2D and 3D graphics and animation 
rendering capabilities.  

 
Illustration 3.1 To illustrate the steps of the procedure, we use the face-colored truncated quadrille 
tiling 𝒬 of ℝଶ, a patch of which is shown in Figure 3.1(a). This tiling is one of the eight semiregular 
Archimedean tilings [9] and consists of congruent regular squares and octagons with one square and 
two octagons arranged around each vertex. We introduce colors to the tiles of 𝒬 so we can easily see 
and analyze the effect of a map on the tiles. Observe that the uncolored tiles of 𝒬 can be generated 
from a single square tile 𝓈 and a single octagon tile ℴ (see Figure 3.1(b)) by successively translating 
them along two linearly independent directions determined by the vectors 𝐯ଵ and 𝐯ଶ. In this case, we 
begin with the tile 𝓈 whose vertices we assign to be (1, 0), (0, 1), (−1, 0), (0, −1). Basic geometry 
and trigonometry computations yield the vertices  

 
൫2 + √2, 1 + √2൯, ൫1 + √2, 2 + √2൯, ൫1, 2 + √2൯, ൫0, 1 + √2൯, (0, 1), (1, 0), ൫1 + √2, 0൯, ൫2 + √2, 1൯ 
 

of the tile ℴ. With these vertices, we deduce the coordinates of the vectors 𝐯ଵ = ൫2 + √2, 0൯ and 

𝐯ଶ = ൫2 + √2, 2 + √2൯. To generate a patch of the uncolored tiling, we translate 𝓈 and ℴ along 
vectors of the form 𝐯 = 𝑐ଵ𝐯ଵ + 𝑐ଶ𝐯ଶ, where the coefficients 𝑐ଵ, 𝑐ଶ are integers at most some pre-
assigned bounding values. The checkerboard coloring of the tiles is achieved by introducing a 
coloring rule based on the parity of 𝑐ଵ. More specifically, for a square or octagon tile, we assign 
yellow or blue, respectively, when 𝑐ଵ is even; or orange or pink, respectively, otherwise. The 
Mathematica implementation of this construction appears in Code 3.1. Note that we utilized the built-
in Polygon function to construct a tile from its list of vertices and the Table function to generate 
translates of 𝓈 and ℴ. We also converted all point and vector coordinates into their numerical or 
decimal equivalent using the function N. This makes computations and graphics rendering in 



Mathematica significantly faster as the system does not need to deal with exact algebraic values 
which often involve radical or trigonometric expressions.  
 

 

 

 

(b) (c) 

  
(a) (d)  (e) 

Figure 3.1 (a) A face-colored tiling 𝒬 of ℝଶ and (b) two tiles used to generate 𝒬 by translations. 
(c – e) Images of the tiling 𝒬𝕋 using the torus map with 𝑎 = ‖6𝐯ଵ‖ and 𝑏 = ‖6(𝐯ଶ − 𝐯ଵ)‖.       

 
yellow = RGBColor[252/256,233/256,79/256,67/100]; 
blue = RGBColor[114/256,159/256,208/256,67/100]; 
orange = RGBColor[245/256,122/256,0/256,67/100]; 
pink = RGBColor[173/256,128/256,168/256,67/100]; 
 
v1 = N[{2+Sqrt[2],0}]; 
v2 = N[{2+Sqrt[2],2+Sqrt[2]}]; 
 
sqVerts = N[{{1,0},{0,1},{-1,0},{0,-1}}]; 
sqPolyYellow = Flatten[Table[Polygon[Table[sqVert+2i*v1+j*v2,{sqVert,sqVerts}]],{i,-4,4},{j,-1,7}],1]; 
sqPolyOrange = Flatten[Table[Polygon[Table[sqVert+(2i+1)*v1+j*v2,{sqVert,sqVerts}]],{i,-4,4},{j,-1,7}],1]; 
 
octVerts = N[{{2+Sqrt[2],1+Sqrt[2]},{1+Sqrt[2],2+Sqrt[2]},{1,2+Sqrt[2]},{0,1+Sqrt[2]},{0,1},{1,0}, 
    {1+ Sqrt[2],0},{2+ Sqrt[2],1}}]; 
octPolyBlue = Flatten[Table[Polygon[Table[octVert+2i*v1+j*v2,{octVert,octVerts}]],{i,-4,4},{j,-1,6}],1]; 
octPolyPink = Flatten[Table[Polygon[Table[octVert+(2i+1)*v1+j*v2,{octVert,octVerts}]],{i,-4,4}, 
    {j,-1,6}],1]; 
 
tiling = Graphics[{EdgeForm[{Black,Thickness[0.005]}],FaceForm[yellow],sqPolyYellow,FaceForm[orange], 
   sqPolyOrange,FaceForm[blue],octPolyBlue,FaceForm[pink],octPolyPink}, 
  PlotRange->{{1.01*Norm[v1],7.01*Norm[v1]},{-0.021Norm[v1],4.23*Norm[v2]}}] 

Code 3.1 Mathematica code to generate the face-colored tiling of 𝒬 in Figure 3.1(a). 
 

Generating an image of a tiling of 𝑆௙ from a tiling 𝒯 then becomes a matter of replacing the 
coordinates of the vertices of the planar tiling with their images under the map 𝑓. Doing this directly, 
however, produces a tiling of 𝑆௙ whose tiles have straight edges rather than smooth curves. To remedy 
this, we first subdivide each edge of 𝒯 into smaller sub-edges by introducing new subdivision points; 
treat the new subdivision points as the new vertices of the tiling; and then apply 𝑓 to these new 
vertices. We discuss this method for the tiling 𝒬 in the next illustration.  
 



Illustration 3.2 We first remark that 𝒬 is an example of a periodic tiling since it possesses both 
horizontal and vertical translation symmetries. Consequently, it is compatible with a torus surface. 
For this illustration, we consider the surface 𝕋 defined by the torus map with 𝑎 ≔ ‖6𝐯ଵ‖ and 𝑏 ≔
‖6(𝐯ଶ − 𝐯ଵ)‖, the length of the red and blue vectors, respectively, in Figure 3.1(a). The additional 
step of subdividing each edge of the tiles in 𝒬 to obtain new vertices is captured in the first five lines 
of Code 3.2. These lines replace the definitions of sqVerts and octVerts in Code 3.1. Here, n 
refers to the number of new subdivision points, including the original endpoints, that will be 
introduced per edge. The greater this number is, the smoother the edges of the resulting tiling 𝒬𝕋 of 
the torus 𝕋 will be. The succeeding lines of code define the torus map; employ this map to obtain the 
vertices of 𝒬𝕋; and render a 3D image 𝒬𝕋. Two perspective views of 𝒬𝕋 are shown in Figure 3.1(c – 
d). Compare these with Figure 3.1(e) corresponding to the case n = 1 in which no new subdivision 
points are introduced to smoothen the resulting bounding curves of the tiles of 𝒬𝕋. We remark that 
Code 3.2 can be easily modified to produce a tiling of another compatible surface by simply replacing 
torusMap with another map.  
 
n = 10; 
 
oldSqVerts = N[{{1,0},{0,1},{-1,0},{0,-1}}]; 
sqVerts = N[Flatten[Table[oldSqVerts[[Mod[k,4]+1]]+(i/n)*(oldSqVerts[[Mod[k+1,4]+1]]-    
           oldSqVerts[[Mod[k,4]+1]]),{k,0,3},{i,0,n-1}],1]]; 
oldOctVerts =  N[{{2+Sqrt[2],1+Sqrt[2]},{1+Sqrt[2],2+Sqrt[2]},{1,2+Sqrt[2]},{0,1+Sqrt[2]},{0,1},{1,0}, 
    {1+Sqrt[2],0},{2+ Sqrt[2],1}}]; 
octVerts = N[Flatten[Table[oldOctVerts[[Mod[k,8]+1]]+(i/n)*(oldOctVerts[[Mod[k+1,8]+1]]-  
            oldOctVerts[[Mod[k,8]+1]]),{k,0,7},{i,0,n-1}],1]]; 
 
torusMap[a_,b_,{u_,v_}] := (1/(Sqrt[a^2+b^2]-a*Cos[u*2π/a]))*{b*Cos[v*2π/b],b*Sin[v*2π/b],a*Sin[u*2π/a]}; 
 
sqPolyYellow = Flatten[Table[Polygon[Table[torusMap[6*Norm[v1],6*Norm[v2-v1],sqVert+2i*v1+j*v2], 
       {sqVert,sqVerts}]],{i,-4,4},{j,-1,7}],1]; 
sqPolyOrange = Flatten[Table[Polygon[Table[torusMap[6*Norm[v1],6*Norm[v2-v1],sqVert+(2i+1)*v1+j*v2], 
       {sqVert,sqVerts}]],{i,-4,4},{j,-1,7}],1]; 
 
octPolyBlue=Flatten[Table[Polygon[Table[torusMap[6*Norm[v1],6*Norm[v2-v1],octVert+2i*v1+j*v2], 
       {octVert,octVerts}]],{i,-4,4},{j,-1,6}],1]; 
octPolyPink=Flatten[Table[Polygon[Table[torusMap[6*Norm[v1],6*Norm[v2-v1],octVert+(2i+1)*v1+j*v2], 
       {octVert,octVerts}]],{i,-4,4},{j,-1,6}],1]; 
 
Graphics3D[{EdgeForm[{Black,Thickness[0.01]}],FaceForm[{yellow,Opacity[0.72], 
     Specularity[yellow, 30]}],sqPolyYellow,FaceForm[{orange,Opacity[0.72],Specularity[orange,30]}], 
  sqPolyOrange,FaceForm[{blue,Opacity[0.72],Specularity[blue,30]}],octPolyBlue, 
  FaceForm[{pink,Opacity[0.72],Specularity[pink, 30]}],octPolyPink},Lighting->"Neutral",Boxed->False, 
 Axes->False] 

Code 3.2 Mathematica code to generate the face-colored tiling of 𝒬𝕋 in Figure 3.1(d – e). 
 

An alternative, more straightforward, and simpler way of generating a tiling of a surface in 
Mathematica is to use the Texture graphics directive within the built-in ParametricPlot3D 
function. The runtime of this second method is significantly faster and its implementation can reduce 
Code 3.2 to just a single line of code! We discuss how this is done in the illustration below. 

 
Illustration 3.3 The second method involves plotting the parametric surface by supplying its 
parametric system of equations and then applying an image of the tiling as texture of the surface. In 
the case of the tiling 𝒬 and the surface 𝕋 in Illustration 3.2, the implementation is found in Code 3.3.  

 
ParametricPlot3D[torusMap[6,6,{u,v}],{u,0,6},{v,0,6},PlotStyle->Directive[Texture[tiling]], 
 Lighting->"Neutral", Mesh-> None,Boxed->False,Axes->False] 

Code 3.3 Mathematica code to generate the face-colored tiling of 𝒬𝕋 in Figure 3.2(a – e). 



The torusMap[6, 6, {u, v}] input in ParametricPlot3D instructs Mathematica to draw the 
parametric surface defined by the torus map introduced earlier in Code 3.2. The bounds 0 and 6 for 
both the parameters u and v indicate that the lightly shaded patch of the tiling in Figure 3.1(a) which 
was supplied as the sole input to Texture must be treated as an image consisting of pixels located 
within a [0, 6] × [0, 6] grid. Mathematica will then automatically perform any rescaling necessary 
for the image. Note that with this given rescaling, both the constants 𝑎 and 𝑏 get the value 6. What 
Mathematica will do essentially is “wrap” the patch around the surface according to the rule defined 
by the supplied parametric equations. That is, it assigns the color or texture of the point (𝑢, 𝑣) to be 
the color or texture of the image point 𝑓(𝑢, 𝑣) in the surface. This method produces the much 
smoother and rounder images of 𝒬𝕋 shown in Figure 3.2(a – d). When the torus map in Code 3.3 is 
replaced by the cylinder map with 𝑎 = 6, we obtain the images for 𝒬ℂ೗

 shown in Figure 3.2(e – g). 
 

 

   
(a) (b) (c) (d) 

   

(e) (f) (g) 
Figure 3.2 Views of the truncated quadrille tilings of the point spaces (a – d) 𝕋 and (e – g) ℂ௟. 

 
We remark that while the second method discussed above is much faster and easier to implement 

than the first, the first provides more flexibility and adaptability when it comes to modifying 
components (vertices, edges, faces) of the tiling separately or independently from one another. This 
is because the second method requires a 2D image of the whole planar tiling to be used as basis for 
texture while the first converts each component of this planar tiling directly into a 3D graphics object. 
Another disadvantage of the second method is that it requires a high-resolution image for texture 
since the map that defines the surface stretches or deforms the image in various directions. 
Minimizing deformations is precisely the reason why we opt to use conformal maps, whenever 
possible, in lieu of ordinary maps.  

 
4.  Rendering Animated Tilings of Surfaces 
 

Now that we have discussed completely how to generate an image of a tiling 𝒯ௌ೑
 of a surface in 

the previous section, we can now demonstrate how to render an animation that creates the effect of 
𝒯 morphing or transforming into 𝒯ௌ೑

. Mathematica offers at least two options for this to be 

accomplished. We can either create an animation with a slider that allows a user to manually control 



the animation or create an animated gif file that automatically runs the animation when opened. In 
either case, we create 𝑚 still frames for the animation where the 𝑖th animation frame, which records 
the 𝑖th state in the entire morphing process, shows the tiling 𝒯 on the parametric surface 

 

൬1 −
𝑖 − 1

𝑚 − 1
൰ ⋅ (𝑢, 𝑣, 0) + ൬

𝑖 − 1

𝑚 − 1
൰ ⋅ ቀ𝑓௫(𝑢, 𝑣), 𝑓௬(𝑢, 𝑣), 𝑓௭(𝑢, 𝑣)ቁ, (4.1) 

 
for 1 ≤ 𝑖 ≤ 𝑚. Observe that when 𝑖 = 1, we just obtain 𝒯 and when 𝑖 = 𝑚, we obtain 𝒯ௌ೑

.  

  
Illustration 4.1 To produce an animation showing 𝒬 morphing into 𝒬𝕋, we run Code 4.1, which 
simply implements the procedure discussed earlier. In this implementation, we created the function 
frame, which generates the 𝑖th animation frame, and used this as an input within Mathematica’s 
Animate function to produce the animation with slider in Figure 4.1(a). The animation list in the 
penultimate line creates a list of eight frames, four of which are shown in Figure 4.1(b – e). When 
this list is called within the Export function, it transforms into an animated gif file which is saved 
in the user’s directory for later retrieval and use. The optional argument DisplayDurations 
specifies the number of seconds delay between showing successive frames.  
 
m = 8; 
 
frame[i_] := ParametricPlot3D[(1-(i-1)/(m-1))*{u,v,0}+((i-1)/(m-1))*torusMap[6,6,u,v],{u,0,6},{v,0,6}, 
   Background->Opacity[0], PlotStyle->Directive[Texture[tiling]],Lighting->"Neutral",Mesh->None, 
   Boxed->False,Axes->False]; 
 
Animate[frame[i],{i,1,m}] 
Animation = Table[frame[i],{i,1,m}]; 
 
Export["animation.gif",animation,"DisplayDurations"->0.5] 

Code 4.1 Mathematica code to render an animation which shows 𝒬 morphing into 𝒬𝕋. 
 
 

 

  
(b) (c) 

  
(a) (d) (e) 

Figure 4.1 (a) Animation with slider showing 𝒬 morphing into 𝒬𝕋. (b – e) Four still frames 
captured from the animation in (a). 

